

## Online Supplementary Information for

Leif Christian Stige, Elena Eriksen, Padmini Dalpadado, Kotaro Ono. Direct and indirect effects of sea ice cover on major zooplankton groups and planktivorous fishes in the Barents Sea. *ICES Journal of Marine Science*.

---

### Contents

---

Appendix S1. Model code	Page S2
Appendix S2. Supplementary Methods	Page S5
Appendix S3. Supplementary tables and figures	Page S9

---

## Appendix S1. Model code.

// The following is the STAN model code

```
data {
  int<lower=0> N; // number of years
  int<lower=0> K; // number of species that are dynamically modelled
  int<lower=0> cop_N; // number of years with copepod data
  int<lower=0> krill_N; // number of years with krill data
  int<lower=0> amph_N; // number of years with amphipod data
  int<lower=0> cap_N; // number of years with capelin data
  int<lower=0> pc_N; // number of years with polar cod data
  int<lower=0> cop_noNA[cop_N]; // rows (years) with copepod data
  int<lower=0> krill_noNA[krill_N]; // rows (years) with krill data
  int<lower=0> amph_noNA[amph_N]; // rows (years) with amph data
  int<lower=0> cap_noNA[cap_N]; // rows (years) with capelin data
  int<lower=0> pc_noNA[pc_N]; // rows (years) with polar cod data
  vector[cop_N] cop_obs; // observed copepod abundance
  vector[krill_N] krill_obs; // observed krill biomass
  vector[amph_N] amph_obs; // observed amphipod biomass
  vector[cap_N] cap_obs; // observed capelin biomass
  vector[pc_N] pc_obs; // observed polar cod biomass
  vector<lower=0>[cop_N] cop_se; // s.e. of cop_obs
  vector<lower=0>[krill_N] krill_se; // s.e. of krill_obs
  vector<lower=0>[amph_N] amph_se; // s.e. of amph_obs
  vector<lower=0>[cap_N] cap_se; // s.e. of cap_obs
  real<lower=0> pc_se; // s.e. of pc_obs
  vector[N] her; // herring biomass
  vector[N] cod; // cod biomass
  vector[N] ice; // ice anomaly dec-may
  vector[N] cap_F; // capelin fishing mortality
}

parameters {
  real c10; // intercept copepods
  real c11; // autoregressive parameter copepods
  real c13; // amphipod effect on copepods
  real c14; // capelin effect on copepods
  real c15; // polar cod effect on copepods
  real c16; // ice effect on copepods

  real c20; // intercept krill
  real c22; // autoregressive parameter krill
  real c24; // capelin effect on krill
  real c26; // ice effect on krill

  real c30; // intercept amphipods
  real c31; // copepod effect on amphipods
  real c33; // autoregressive parameter amphipods
  real c35; // polar cod effect on amphipods
  real c36; // ice effect on amphipods

  real c40; // intercept capelin
  real c41; // copepod effect on capelin
  real c42; // krill effect on capelin
  real c44; // autoregressive parameter capelin
  real c46; // ice effect on capelin
  real c47; // cod effect on capelin
  real c48; // herring effect on capelin
  real c49; // fishing on capelin

  real c50; // intercept polar cod
  real c51; // copepod effect on polar cod
  real c53; // amphipod effect on polar cod
  real c55; // autoregressive parameter polar cod
  real c56; // ice effect on polar cod
  real c57; // cod effect on polar cod
```

```

real<lower=0.8> s1o; // multiplication factor observation error copepods
real<lower=0.8> s2o; // multiplication factor observation error krill
real<lower=0.8> s3o; // multiplication factor observation error amphipods
real<lower=0.8> s4o; // multiplication factor observation error capelin
real<lower=0.5> s5o; // multiplication factor observation error polar cod

// covariance matrix process errors of (cop,krill,amph,cap,pc)
cholesky_factor_corr[K] L_Omega;
vector<lower=0.1>[K] L_sigma;

// latent values of (cop,krill,amph,cap,pc)
matrix<lower = -4, upper = 4>[N,K] Latent;
}

model {
  matrix[K,K] L_Sigma;

  // prior distribution of parameters
  c10 ~ cauchy(0, 5);
  c11 ~ cauchy(0, 5);
  c13 ~ cauchy(0, 5);
  c14 ~ cauchy(0, 5);
  c15 ~ cauchy(0, 5);
  c16 ~ cauchy(0, 5);
  c20 ~ cauchy(0, 5);
  c22 ~ cauchy(0, 5);
  c24 ~ cauchy(0, 5);
  c26 ~ cauchy(0, 5);
  c30 ~ cauchy(0, 5);
  c31 ~ cauchy(0, 5);
  c33 ~ cauchy(0, 5);
  c35 ~ cauchy(0, 5);
  c36 ~ cauchy(0, 5);
  c40 ~ cauchy(0, 5);
  c41 ~ cauchy(0, 5);
  c42 ~ cauchy(0, 5);
  c44 ~ cauchy(0, 5);
  c46 ~ cauchy(0, 5);
  c47 ~ cauchy(0, 5);
  c48 ~ cauchy(0, 5);
  c49 ~ normal(-1, .2);
  c50 ~ cauchy(0, 5);
  c51 ~ cauchy(0, 5);
  c53 ~ cauchy(0, 5);
  c55 ~ cauchy(0, 5);
  c56 ~ cauchy(0, 5);
  c57 ~ cauchy(0, 5);

  s1o ~ normal(1.2, .2);
  s2o ~ normal(1.2, .2);
  s3o ~ normal(1.2, .2);
  s4o ~ normal(1.2, .2);
  s5o ~ normal(1, .4);

  L_Sigma = diag_pre_multiply(L_sigma, L_Omega);
  L_Omega ~ lkj_corr_cholesky(4);
  L_sigma ~ normal(0, 5);

  // process model
  Latent[1,1] ~ normal(0, 3);
  Latent[1,2] ~ normal(0, 3);
  Latent[1,3] ~ normal(0, 3);
  Latent[1,4] ~ normal(0, 3);
  Latent[1,5] ~ normal(0, 3);
  for(n in 2:N){
    // Expected:
    vector[K] Mu;
    //cop:

```

```

Mu[1] = c10
      + c11 * Latent[n-1, 1]
      + c13 * Latent[n-1, 3]
      + c14 * Latent[n-1, 4]
      + c15 * Latent[n-1, 5]
      + c16 * ice[n];
// krill:
Mu[2] = c20
      + c22 * Latent[n-1, 2]
      + c24 * Latent[n-1, 4]
      + c26 * ice[n];
// amph:
Mu[3] = c30
      + c31 * Latent[n-1, 1]
      + c33 * Latent[n-1, 3]
      + c35 * Latent[n-1, 5]
      + c36 * ice[n];
// cap:
Mu[4] = c40
      + c41 * Latent[n-1, 1]
      + c42 * Latent[n-1, 2]
      + c44 * Latent[n-1, 4]
      + c46 * ice[n]
      + c47 * cod[n]
      + c48 * her[n]
      + c49 * cap_F[n] ;
// pc:
Mu[5] = c50
      + c51 * Latent[n-1, 1]
      + c53 * Latent[n-1, 3]
      + c55 * Latent[n-1, 5]
      + c56 * ice[n]
      + c57 * cod[n] ;
// Latent = Expected + process error:
Latent[n,] ~ multi_normal_cholesky(Mu, L_Sigma);
}

// observation model:
cop_obs ~ normal(Latent[cop_noNA,1], s1o * cop_se);
krill_obs ~ normal(Latent[krill_noNA,2], s2o * krill_se);
amph_obs ~ normal(Latent[amph_noNA,3], s3o * amph_se);
cap_obs ~ normal(Latent[cap_noNA,4], s4o * cap_se);
pc_obs ~ normal(Latent[pc_noNA,5], s5o * pc_se);
}

generated quantities {
corr_matrix[K] Omega;
Omega = multiply_lower_tri_self_transpose(L_Omega);
}

```

## Appendix S2. Supplementary Methods.

### *Fitting the model*

All data series were normalised to zero mean and unit standard deviation prior to analysis to facilitate convergence. For ease of interpretation, the estimated parameters and state variables were transformed back to the original (i.e., natural logarithm) scale of the data when presenting model output. To further facilitate convergence, we used Cholesky factorisation of the variance-covariance matrix  $\Sigma$ . That is, we parameterized the model so that we estimated a scale vector  $\tau_1, \dots, \tau_5$  and the Cholesky factor of the correlation matrix  $\Omega$ , which was inverse transformed to  $\Omega$  after estimation (Stan Development Team, 2018). The squared elements of the scale vector ( $\tau_1^2, \dots, \tau_5^2$ ) represented the variance components of  $\Sigma$ . The Stan model code is given in the online Appendix S1.

We used four independent chains with 20 000 iterations, where the first 10 000 iterations were used as ‘warm-up’ iterations. In addition, we thinned the chains with a factor 20 to reduce autocorrelation in the posterior samples and to produce a reasonable amount of output. In the end, we had 500 samples from each chain, leading to a total of 2000 samples. To avoid divergent transitions in the sampling of the posterior distribution (Stan Development Team, 2018), the ‘adapt\_delta’ parameter was set to 0.995 and maximum tree depth to 15.

### *Prior distributions*

We chose non-informative priors for most model parameters to let the data drive the inferences. Specifically, Cauchy distributions with means zero and standard deviations 5 were used as priors for all parameters  $c_{ij}$  except for  $c_{49}$ , i.e., the coefficient for the effect of fishing,  $F_{CAP}$ . The variable  $F_{CAP}$  was defined so that we expected  $c_{49}$  to be close to -1, but with uncertainty due to uncertainty and trend in population biomass during the period of fishing and uncertainty in fisheries’ catch. Hence, we used a normal distribution with mean -1 and standard deviation 0.2 as prior for  $c_{49}$ .

We used informative priors for the standard deviations for the observation noise  $\sigma_{1,T}, \dots, \sigma_{5,T}$ . For the first four time-series, *Cop*, *Krill*, *Amph* and *Cap*, the priors were based on annual nominal standard errors,  $s_{Cop,T}$ ,  $s_{Krill,T}$ ,  $s_{Amph,T}$ ,  $s_{Cap,T}$  (calculated as described in *Appendix S2. Supplementary Methods*). The annual standard deviations  $\sigma_{1,T}, \dots, \sigma_{4,T}$  for *Cop*, *Krill*, *Amph* and *Cap* were assumed to scale with the respective annual nominal standard errors, so that we only estimated term-specific scaling factors  $\sigma_1, \dots, \sigma_4$ . Specifically, we assumed  $\sigma_{1,T} = \sigma_1 s_{Cop,T}$ ,  $\sigma_{2,T} = \sigma_2 s_{Krill,T}$ ,  $\sigma_{3,T} = \sigma_3 s_{Amph,T}$ , and  $\sigma_{4,T} = \sigma_4 s_{Cap,T}$ . The prior for each scaling factor was a truncated normal distribution centred around 1.2 (to account for underestimation of uncertainty by the nominal standard errors) with a standard deviation of 0.2 and a lower bound of 0.8. While we lacked quantitative information about the

uncertainty of *Pol*, polar cod and capelin were estimated acoustically at the same surveys, so many uncertainty sources were similar. The survey mainly targeted the commercially important capelin however, and the distribution area of polar cod was not fully covered in all years. We therefore expected the uncertainty of *Pol* to be of the same order of magnitude as that of *Cap*, but somewhat higher. For *Pol* we assumed the same observation error for all years,  $\sigma_{5,T} = \sigma_5$ , with the prior for  $\sigma_5$  being a truncated normal distribution with mean 0.3, standard deviation 0.12 and a lower bound 0.15, i.e. with both higher mean and higher standard error than capelin for most years (as the median nominal standard error for capelin was 0.14).

The priors for the scales of the process noise,  $\tau_1, \dots, \tau_5$ , were also assumed to follow a truncated normal distribution with mean zero, standard deviation 5 and lower bound 0.1. As the priors referred to the normalized scale the model was fitted to, these lower bounds meant that process noise was assumed to contribute at least 1 % of the observed variance. We used an LKJ prior with shape 4 for the Cholesky factor of the correlation matrix  $\Omega$ . With a shape parameter of 1, the LKJ correlation distribution reduces to the uniform distribution over correlation matrices of order 5; as the shape parameter increases, the prior increasingly concentrates around the unit correlation matrix (Stan Development Team, 2018). These restrictions on the process error variance-covariance (i.e., lower bounds on scale and shape parameter larger than 1) were applied due to difficulties in sampling unrealistically low or strongly correlated process errors (causing divergent transitions, Stan Development Team, 2018).

The priors for the initial values of the state variables  $x_{1,T}, \dots, x_{5,T}$  at time  $T = 1$  were normal distributions with means zero and standard deviations 3, and the state variables were bounded to be between -4 and 4 at all time steps (at the normalised scale the model was fitted in).

### *Uncertainty estimates*

Information about the uncertainty of the time-series was used as input to the analysis (as described in the section *Prior distributions*).

Annual nominal standard errors ( $s_{Cop,T}$ ) of *Cop* were calculated by averaging the nominal standard errors of the regional abundance indices ( $s_{CopC,T}$  and  $s_{CopN,T}$ , respectively, obtained from the regression models used to calculate the logarithmic scale indices *Cop<sub>C</sub>* and *Cop<sub>N</sub>*, see Stige *et al.* (2014)). Specifically, we used  $s_{Cop,T} = (s_{CopC,T}^2 + s_{CopN,T}^2)^{0.5} / 2$ . Median and range of  $s_{Cop,T}$  were 0.06 and (0.05, 0.14), contributing to around 4 % of the observed variance in *Cop*.

Annual nominal standard errors ( $s_{Krill,T}$ ) of *Krill* were calculated from nominal standard errors  $s_{Krill,D,T}$  and  $s_{Krill,N,T}$  of the arithmetic scale indices *Krill<sub>D</sub>* and *Krill<sub>N</sub>*. The standard errors  $s_{Krill,D,T}$  and  $s_{Krill,N,T}$

were calculated from the between-station standard deviation ( $sd$ ) and number of stations ( $n$ ) for each sampling period (day or night) and year using  $s = sd / n^{0.5}$ . A joint arithmetic scale coefficient of variation was calculated as  $CV_{Krill,T} = (s_{Krill,D,T}^2 + s_{Krill,N,T}^2)^{0.5} / ([Krill_D + Krill_N])$ . The logarithmic scale standard error ( $s_{Krill,T}$ ) was calculated from  $CV_{Krill,T}$  using  $s = \ln(1 + CV^2)^{0.5}$ . Median and range of  $s_{Krill,T}$  were 0.43 and (0.27, 0.78), contributing to around 12 % of the observed variance in *Krill*.

Annual nominal standard errors  $s_{Amph,T}$  of *Amph* were obtained from the regression model used to calculate *Amph* (Dalpadado *et al.*, 2012). Median and range of  $s_{Amph,T}$  were 0.12 and (0.08, 0.29), contributing to around 5 % of the observed variance in *Amph*.

To calculate annual nominal standard errors  $s_{Cap,T}$  of *Cap*, we first calculated the arithmetic scale coefficients of variation for biomass at ages  $a = 1-4$ ,  $CV_{CapBM,a,T} = (\exp[s_{CapN,a,T}^2 + s_{CapW,a,T}^2] - 1)^{0.5}$ . Here,  $s_{CapN,a,T}$  is the standard error of log-abundance-at-age,  $s_{CapW,a,T}$  is the standard error of log-weight-at-age, and  $s_{CapN,a,T}^2 + s_{CapW,a,T}^2$  is the squared standard error of log-biomass-at-age (for simplicity assuming independent observation errors of abundance and weight). Estimates of  $s_{CapN,a,T}$  were mainly based on Tjelmeland (Tjelmeland, 2002; see Stige *et al.*, 2018 for details), i.e., 0.2 for most years but 0.5 or 1 for some ages in some years. We used  $s_{CapW,a,T} = s_{CapW} = 0.06$  for all ages and years, based on an estimated standard error of log-length-at-age around 0.02 (Stige *et al.*, 2018) and an approximately cubic relationship between length and weight (corresponding to a multiplication factor of 3 on log-scale). The coefficient of variation for total biomass of ages 1–4 was calculated as  $CV_{CapBM,T} = [\sum_{a=1}^{a=4} (CV_{CapBM,a,T}^2 BM_{Cap,a,T}^2)]^{0.5} / \sum_{a=1}^{a=4} (BM_{Cap,a,T})$ , based on the simplifying assumption that observation errors for different ages are independent. Annual nominal standard errors  $s = s_{Cap,T}$  were calculated from  $CV = CV_{CapBM,T}$  using  $s = \ln(1 + CV^2)^{0.5}$ . Median and range of  $s_{Cap,T}$  were 0.14 and (0.12, 0.38), contributing to around 2 % of the observed variance in *Cop*.

Note that these nominal standard errors are likely to underestimate uncertainty, for example due to unaccounted spatial correlations in the construction of zooplankton indices, and across-age-correlations for capelin. For capelin, our data allows assessing the potential magnitude of this effect: If observation noise is correlated across capelin ages so that the uncertainty around the total biomass is equal to the uncertainty around a single age-class, the standard error for capelin becomes approximately 50 % higher than the nominal values calculated above (with a median value of 0.21 instead of 0.14).

We lacked quantitative information about the uncertainty of *Pol* and the other time-series analysed.

## References

- Coyle, K. O., and Gibson, G. A. 2017. *Calanus* on the Bering Sea shelf: probable cause for population declines during warm years. *Journal of Plankton Research*, 39: 257-270.
- Dalpadado, P., Ingvaldsen, R. B., Stige, L. C., Bogstad, B., Knutsen, T., Ottersen, G., and Ellertsen, B. 2012. Climate effects on the Barents Sea ecosystem dynamics. *ICES Journal of Marine Science*, 69: 1303-1316.
- ICES 2018. Interim Report of the Working Group on the Integrated Assessments of the Barents Sea (WGIBAR), 9-12 March 2018, Tromsø, Norway, ICES CM 2018/IEASG:04. 210 pp.
- Stan Development Team 2018. *Stan Modeling Language Users Guide and Reference Manual*, Version 2.18.0, <http://mc-stan.org>. <http://mc-stan.org> pp.
- Stige, L. C., Dalpadado, P., Orlova, E., Boulay, A.-C., Durant, J. M., Ottersen, G., and Stenseth, N. C. 2014. Spatiotemporal statistical analyses reveal predator-driven zooplankton fluctuations in the Barents Sea. *Progress in Oceanography*, 120: 243-253.
- Stige, L. C., Kvile, K. Ø., Bogstad, B., and Langangen, Ø. 2018. Predator-prey interactions cause apparent competition between marine zooplankton groups. *Ecology*, 99: 632-641.
- Tjelmeland, S. 2002. A model for the uncertainty around the yearly trawl-acoustic estimate of biomass of Barents Sea capelin, *Mallotus villosus* (Müller). *ICES Journal of Marine Science*, 59: 1072-1080.
- Trofimov, A., and Ingvaldsen, R. 2016. Chapter 3.1 Hydrography. *In* Survey report from the joint Norwegian/Russian ecosystem survey in the Barents Sea and adjacent waters, August-October 2015. Ed. by D. Prozorkevich, and K. Sunnannå. IMR/PINRO Joint Report series, No. 1/2016. ISSN 1502-8828.
- Wassmann, P., Reigstad, M., Haug, T., Rudels, B., Carroll, M. L., Hop, H., Gabrielsen, G. W., et al. 2006. Food webs and carbon flux in the Barents Sea. *Progress in Oceanography*, 71: 232-287.



### Appendix S3. Supplementary tables and figures.

**Table S1.** Alternative climate variables. In a series of alternative models, winter sea ice (*Ice*), used as climate variable in the main analysis, was replaced by each of the alternative climate variables.

Climate variable	Description
Sea ice cover in April	Area of sea ice in April, which is the month of maximal ice coverage (ICES, 2018). The index was calculated based on sea ice concentration data from the National Snow and Ice Data Center for 72–82 °N, 10–60 °E.
Area of Arctic water masses in autumn	Area of Arctic water masses (temperature < 0 °C) in the area 72–80°N, 20–50°E (ICES, 2018). The index was calculated based on the mean 50–200 m temperature fields from temperature measurements taken during annual scientific surveys in the third quarter. The index has been shown to be positively associated with the occurrence of pelagic amphipods in the Barent Sea (Dalpadado <i>et al.</i> , 2012).
Summer SST	Annual mean summer (May–September) sea surface temperature (SST) in the area 74–80 °N, 20–50 °E (Fig. 1). This period covers the main primary and secondary production season in the central and northern Barents Sea (Wassmann <i>et al.</i> , 2006). The index was calculated from monthly sea surface temperature data at a 2×2° grid (NOAA_ERSST_V3 data set) provided by the NOAA/OAR/ESRL PSD, Boulder, Colorado, USA, from their Web site at <a href="http://www.esrl.noaa.gov/psd/">http://www.esrl.noaa.gov/psd/</a> .
Spring SST	SST in May–June, calculated from the same data as Summer SST. SST is hypothetically most critical early in the season, e.g. through association with the timing of the spring bloom.
Previous autumn SST	Previous year’s SST in August–November, calculated from the same data as Summer SST. Previous-autumn temperature has been negatively associated with variation in copepod biomass in the central and northern Barents Sea (Stige <i>et al.</i> , 2014).
Previous autumn water column temperature	Previous year’s water column temperature in August–September, measured at 50–200 m depth at the Vardø North section at 72°15′N – 74°15′N, 31°13′E (Trofimov and Ingvaldsen, 2016, with missing value for 2010 interpolated between values for 2009 and 2011). High water column temperature in autumn is hypothetically a proxy for high overwintering temperature and -metabolism, which have been suggested to negatively affect copepod survival in the Bering Sea (Coyle and Gibson, 2017).

**Table S2.** Prior and posterior distributions of model parameters. Values are at the normalised scale the model was fitted to†. Numbers in bold: 95 % credibility intervals that exclude zero.


Parameter	Prior Mean	Prior SD	Posterior median	Posterior mean	Posterior SD	Posterior 2.5% quantile	Posterior 97.5% quantile
<i>c10</i>	0	5	0.02	0.03	0.14	-0.25	0.30
<i>c11</i>	0	5	0.05	0.00	0.37	-0.81	0.64
<i>c13</i>	0	5	0.00	-0.02	0.35	-0.71	0.63
<i>c14</i>	0	5	-0.54	-0.57	0.28	<b>-1.17</b>	<b>-0.04</b>
<i>c15</i>	0	5	0.20	0.21	0.31	-0.33	0.89
<i>c16</i>	0	5	0.40	0.39	0.19	<b>0.01</b>	<b>0.75</b>
<i>c20</i>	0	5	0.05	0.05	0.11	-0.17	0.26
<i>c22</i>	0	5	0.59	0.59	0.17	<b>0.25</b>	<b>0.91</b>
<i>c24</i>	0	5	-0.29	-0.29	0.15	<b>-0.58</b>	<b>-0.01</b>
<i>c26</i>	0	5	-0.18	-0.18	0.13	-0.42	0.09
<i>c30</i>	0	5	0.04	0.04	0.14	-0.22	0.33
<i>c31</i>	0	5	0.40	0.39	0.23	-0.06	0.84
<i>c33</i>	0	5	-0.07	-0.08	0.28	-0.65	0.47
<i>c35</i>	0	5	-0.58	-0.60	0.30	<b>-1.32</b>	<b>-0.05</b>
<i>c36</i>	0	5	0.40	0.40	0.18	<b>0.06</b>	<b>0.78</b>
<i>c40</i>	0	5	-0.04	-0.04	0.08	-0.20	0.12
<i>c41</i>	0	5	-0.01	-0.01	0.19	-0.41	0.35
<i>c42</i>	0	5	0.74	0.74	0.23	<b>0.29</b>	<b>1.19</b>
<i>c44</i>	0	5	1.04	1.04	0.15	<b>0.72</b>	<b>1.33</b>
<i>c46</i>	0	5	-0.26	-0.26	0.12	<b>-0.48</b>	<b>-0.01</b>
<i>c47</i>	0	5	-0.60	-0.59	0.20	<b>-0.98</b>	<b>-0.20</b>
<i>c48</i>	0	5	-0.21	-0.20	0.12	-0.44	0.04
<i>c49</i>	-1	0.2	-1.05	-1.06	0.20	<b>-1.47</b>	<b>-0.69</b>
<i>c50</i>	0	5	-0.03	-0.03	0.13	-0.30	0.24
<i>c51</i>	0	5	0.21	0.22	0.30	-0.41	0.82
<i>c53</i>	0	5	-0.09	-0.07	0.41	-0.87	0.76
<i>c55</i>	0	5	0.71	0.73	0.31	<b>0.16</b>	<b>1.38</b>
<i>c56</i>	0	5	-0.11	-0.12	0.22	-0.59	0.27
<i>c57</i>	0	5	-0.06	-0.08	0.27	-0.69	0.40
$\sigma_1$	1.2	0.2	1.22	1.23	0.20	<b>0.87</b>	<b>1.63</b>
$\sigma_2$	1.2	0.2	1.26	1.25	0.19	<b>0.88</b>	<b>1.60</b>
$\sigma_3$	1.2	0.2	1.23	1.24	0.19	<b>0.88</b>	<b>1.63</b>
$\sigma_4$	1.2	0.2	1.14	1.16	0.18	<b>0.84</b>	<b>1.54</b>
$\sigma_5$	0.3	0.12	0.41	0.40	0.10	<b>0.19</b>	<b>0.58</b>
$\tau_1$	‡	5	0.65	0.66	0.13	<b>0.41</b>	<b>0.94</b>
$\tau_2$	‡	5	0.57	0.59	0.12	<b>0.38</b>	<b>0.85</b>
$\tau_3$	‡	5	0.46	0.46	0.16	<b>0.17</b>	<b>0.77</b>
$\tau_4$	‡	5	0.30	0.30	0.11	<b>0.11</b>	<b>0.53</b>
$\tau_5$	‡	5	0.57	0.58	0.22	<b>0.17</b>	<b>1.02</b>
$\Omega[1,2]$	0	§	0.37	0.35	0.23	-0.14	0.74
$\Omega[1,3]$	0	§	0.26	0.24	0.24	-0.26	0.67
$\Omega[1,4]$	0	§	-0.15	-0.14	0.26	-0.62	0.39
$\Omega[1,5]$	0	§	0.15	0.14	0.25	-0.34	0.62

$\Omega[2,3]$	0	§	0.16	0.16	0.25	-0.35	0.60
$\Omega[2,4]$	0	§	-0.20	-0.19	0.26	-0.67	0.35
$\Omega[2,5]$	0	§	0.04	0.04	0.25	-0.45	0.50
$\Omega[3,4]$	0	§	-0.21	-0.19	0.26	-0.65	0.35
$\Omega[3,5]$	0	§	-0.05	-0.05	0.29	-0.62	0.50
$\Omega[4,5]$	0	§	0.08	0.07	0.26	-0.45	0.55


† The coefficients can here be interpreted as standard deviation unit change in response variable per standard deviation unit change in predictor variable (except the coefficient for  $F_{CAP}$ , which was scaled to be  $-1$  if reported catch and stock biomasses were correct and natural mortality during the period of fishing was ignored). The standard deviations of *Cop*, *Krill*, *Amph*, *Cap*, *Pol*, *Ice*, *Cod* and *Her* were, respectively, 0.364, 1.43, 0.613, 1.20, 0.845, 14.9, 0.563 and 1.26. These standard deviations allowed converting coefficient between normalised scale (this table) and the scale of the data (Fig. 4). Hence, for example, the ln-scale change in copepods per ln-scale change in capelin ( $C_{14}$ ) was  $-0.54 \cdot 0.364 / 1.20$ . The standard deviation of  $F_{CAP}$  was 0.11, meaning that the standard deviation unit change in *Cap* per standard deviation unit change in  $F_{CAP}$  was  $-1 \cdot 0.11$ .

‡ The priors for the scales of the process noise,  $\tau_1, \dots, \tau_5$ , were truncated normal distributions with means zero, standard deviations 5 and lower bounds 0.1.


§ We used an LKJ prior with shape 4 for the Cholesky factor of the correlation matrix  $\Omega$ .




**Figure S1.** Alternative climate variables. Sea ice cover in winter (*Ice*) was the climate variable used in the main analysis. The alternative climate variables were sea ice cover in April, area of Arctic water masses in autumn, summer sea surface temperature, spring sea surface temperature, previous-autumn sea surface temperature, and previous-autumn water column temperature (see Table S1 for definition of variables).




**Figure S2.** Log-biomass at year  $T$  versus log-biomass at year  $T-1$  for the biological time-series in focus. The relationship should be linear if the Gompertz model is appropriate. As we saw no strong indications of non-linearity, we considered this assumption reasonable.



**Figure S3.** Estimated model coefficients at the normalised scale the model was fitted to. The coefficients can here be interpreted as standard deviation unit change in response variable per standard deviation unit change in predictor variable (except the coefficient for  $F_{CAP}$ , which is scaled to be -1 if reported catch and stock biomasses are correct and natural mortality during the period of fishing is ignored).




**Figure S4.** Correlation in parameter estimates. The figure shows pairwise associations between posterior samples of model coefficients that had product-moment correlation coefficient ( $r$ ) above 0.4 in absolute value. Dashed lines show locations of zero (no effect). Several of the coefficients were correlated, meaning that they were best interpreted pairwise. For example, the density dependence parameter for copepods ( $c_{11}$ ) was correlated with several other parameters. We used a broad prior distribution for  $c_{11}$  and the posterior distribution included both positive and negative values of  $c_{11}$ . Negative values of  $c_{ij}$  imply overcompensation, meaning that high biomass concentrations one year led to low values the next year, for example due to overexploitation of resources at high biomass or cannibalism. If we considered that overcompensation in copepods was unlikely and only considered posterior samples with  $c_{11} > 0$ , the posterior distributions of the correlated parameters changed: the effect of amphipods on copepods ( $c_{13}$ ) was more likely negative, the effect of capelin on copepods ( $c_{14}$ ) was more weakly negative and the effect of polar cod on copepods ( $c_{15}$ ) was closer to zero compared to the case where such restriction on  $c_{11}$  was not made. Similarly, if we considered that overcompensation in amphipods was unlikely and assumed that  $c_{33} > 0$ , the effect of copepods on amphipods ( $c_{31}$ ) was more weakly positive and the effect of polar cod on amphipods ( $c_{35}$ ) was more weakly negative.




**Figure S5.** Estimated model coefficients for alternative model formulations. Filled black circles and bars show posterior means and 95 % credibility intervals of the main model also shown in **Fig. 4**. Blue open circles and bars show results of a model that includes direct interactions between polar cod and krill (parameters  $C_{25}$  and  $C_{52}$ ). Green squares and bars show results of a model that includes direct interactions between copepods and krill (parameters  $C_{12}$  and  $C_{21}$ ). The alternative models show no strong interactions between polar cod and krill or between copepods and krill (95 % credibility intervals include zero). The other parameters are not very sensitive to these alterations in model formulation, but 95 % credibility intervals are generally broader in the alternative models.





**Figure S6.** Estimated coefficients for models with alternative climate variables (listed in Table S1). Points and bars show posterior means and 95 % credibility intervals at the normalised scale the model was fitted to (as in Fig. S3). Blue filled circles (and bars) show results for the baseline model with *Ice* as climate variable. In the alternative models, *Ice* was replaced by: sea ice cover in April (filled upward-pointing triangles), area of Arctic water masses in autumn (filled squares), summer sea surface temperature (open circles), spring sea surface temperature (open upward-pointing triangles), previous-autumn sea surface temperature (open downward-pointing triangles), or previous-autumn water column temperature (open squares).



**Figure S7.** Estimated direct and indirect effects of climate for models with alternative climate variables (listed in Table S1). Points and bars show posterior means and 95 % credibility intervals for the standard deviation unit change in the response variable per standard deviation unit change in the climate variable. The direct effects are the estimated effects of a given climate variable on each organism. The indirect effects are the estimated effects of the climate variable through the other analysed organisms and occur with a 1-year time lag. Blue filled circles (and bars) show results for the baseline model with *Ice* as climate variable. In the alternative models, *Ice* was replaced by: sea ice cover in April (filled upward-pointing triangles), area of Arctic water masses in autumn (filled squares), summer sea surface temperature (open circles), spring sea surface temperature (open upward-pointing triangles), previous-autumn sea surface temperature (open downward-pointing triangles), or previous-autumn water column temperature (open squares).